

第 3 章

汇编语言程序设计实验

3.1 顺序程序设计

实验目的：

- (1) 掌握顺序程序设计方法。
- (2) 学习数据传送及算术和逻辑运算指令的用法。
- (3) 熟悉在 PC 上建立、汇编、连接、调试和运行 8086 汇编语言程序的过程。

[程序 1] 设 X 和 Y 均为 16 位无符号数,写一个求表达式 $X+Y$ 值的程序。

程序清单如下：

```
DSEG    SEGMENT                                ;数据段
X        DW        1234H
Y        DW        5678H
Z        DW        ?
DSEG    ENDS
CSEG    SEGMENT                                ;代码段
        ASSUME     CS:CSEG,DS:DSEG
START:  MOV        AX,DSEG                    ;段寄存器初值
        MOV        DS,AX
        MOV        AX,X
        ADD        AX,Y                      ;求 X+Y 和
        MOV        Z,AX                      ;保存
        MOV        AH,4CH
        INT        21H                      ;程序结束退出
CSEG    ENDS
        END        START
```

思考题：本程序没有考虑进位的情况。如果考虑可能产生的进位,应如何修改程序？

扩展题：设 X 和 Y 均为三字节无符号数,编制程序求其和,并将结果放置在 Z 开始的存储单元。

[程序 2] 编制程序,将字节变量 BVAR 中的压缩 BCD 数转换成二进制数,并存入

原变量中。

在编写程序前,首先要确定转换算法。第一步,取压缩 BCD 数的十位;第二步,计算十位数乘以 10;第三步,取个位数相加。

程序清单如下所示:

```
DSEG  SEGMENT
BVAR  DB 99H
DSEG  ENDS
CSEG  SEGMENT
      ASSUME CS:CSEG,DS:DSEG
START: MOV     AX,DSEG
      MOV     DS,AX
      MOV     AL,BVAR
      MOV     CL,4
      SHR     AL,CL           ;取高 4 位
      MOV     AH,10
      MUL     AH             ;乘 10
      AND     BVAR,0FH       ;取低 4 位
      ADD     BVAR,AL        ;相加
      MOV     AH,4CH
      INT     21H
CSEG  ENDS
      END     START
```

思考题: 程序中求个位数乘以 10 采用乘法指令 MUL 实现,而用移位指令也可以实现,请问该如何修改程序?

扩展题: 这个程序还有另外一个编制算法,即首先将压缩 BCD 数转换成非压缩 BCD 数,然后通过 BCD 码调整指令将其直接转换成二进制数。请问应该使用哪一个调整指令? 试编制程序。

[程序 3] 内存中自 TABLE 开始的 10 个单元连续存放着 0 到 9 的平方值(称为平方表)。从键盘输入一位十进制数 $X(0 \leq X \leq 9)$,查表求 X 的平方值,结果存入 Y 单元,并将该数输出到屏幕上。

程序清单如下:

```
DATA  SEGMENT
TABLE DB      0,1,4,9,16,25,36,49,64,81    ;定义平方表
Y      DB      ?
DATA  ENDS
CODE  SEGMENT
      ASSUME CS:CODE,DS:DATA
START:
      MOV     AX,DATA
      MOV     DS,AX                ;置数据段寄存器
```

```

LEA      BX, TABLE
MOV      AH, 1
INT      21H                      ;DOS 功能调用的 1 号子功能是键盘输入
SUB      AL, 30H                  ;返回值为 ASCII 码,存于 AL 中
XLAT     ;AL←((BX)+(AL))
MOV      Y, AL                    ;平方数存 Y 单元
MOV      AH, 2                    ;DOS 功能调用的 2 号子功能是屏幕输出
MOV      DL, 09H                  ;输出一个 Tab
INT      21H
MOV      AL, Y                    ;取平方值除以 10,商为十位数,余数为个位数
XOR      AH, AH
MOV      BL, 10
DIV      BL
MOV      BL, AH
ADD      AL, 30H                  ;十位数转换成 ASCII 码输出
MOV      DL, AL
MOV      AH, 2
INT      21H
ADD      BL, 30H                  ;个位数转换成 ASCII 码输出
MOV      DL, BL
INT      21H
MOV      AH, 4CH                  ;返回 DOS
INT      21H
CODE     ENDS
END      START

```

思考题：如果题目要求计算 0~20 之间的数的平方值，还能采用查表指令 XLAT 实现吗？为什么？

扩展题：自 TABLE 开始的 10 个单元连续存放着 2 的 0 次方到 2 的 9 次方的值。依然是从键盘输入一位十进制数 $X(0 \leq X \leq 9)$ ，求 2 的 X 次方值，并把结果存入 Y 单元。

3.2 分支程序设计

实验目的：

- (1) 掌握分支程序的结构。
- (2) 掌握分支程序的设计，调试方法。

〔程序 1〕 设有 3 个单字节无符号数存放在 BUF 开始的缓冲区中，编写一个能将它们从大到小重新排序的程序。

由于 BUF 缓冲区中只有 3 个数据，有多种方法可实现 3 个数据的排序。在此采用交换法，先找到 3 个数中的最大数，然后再找到剩下两个数的大数，最后将 3 个数据按大小依次存放。为了方便，先把要排序的 3 个数取到 3 个寄存器中，然后再对 3 个数进行比较

排序。编写的源程序如下所示：

```
DATA    SEGMENT
BUF      DB      87,234,123
DATA    ENDS
CODE    SEGMENT
        ASSUME   CS:CODE,DS:DATA
START:  MOV      AX,DATA
        MOV      DS,AX
        MOV      SI,OFFSET BUF
        MOV      AL,[SI]           ;把 3 个数取到寄存器中
        MOV      BL,[SI+1]
        MOV      CL,[SI+2]
        CMP      AL,BL           ;排序,将最大数送 AL 寄存器
        JAE      NEXT1
        XCHG     AL,BL
NEXT1:  CMP      AL,CL
        JAE      NEXT2
        XCHG     AL,CL
NEXT2:  CMP      BL,CL           ;将最小数送 CL 寄存器
        JAE      NEXT3
        XCHG     BL,CL
NEXT3:  MOV      [SI],AL         ;从大到小依次存回缓冲区
        MOV      [SI+1],BL
        MOV      [SI+2],CL
        MOV      AH,4CH
        INT      21H
CODE    ENDS
        END      START
```

思考题：分支结构程序设计的关键在于准确地知道操作结果影响的标志位状态和正确地使用条件转移指令。本程序中，由于是无符号数的比较，所以也可以根据一个标志位来判定数的大小，你能说出是哪个标志位吗？程序中的 JAE 指令该用什么指令替换呢？

扩展题：在 BUF 开始的单元里有 3 个单字节无符号数，编写程序找出它们中的最小值并存入 MIN 单元。

〔程序 2〕 编写一个程序，判别键盘上输入的字符；若是 0~9 字符，则显示之；若为 A~Z 或 a~z 字符，均显示“C”；若是回车字符<CR>（其 ASCII 码为 0DH），则结束程序，若为其他字符则不显示，继续等待新的字符输入。

程序清单如下：

```
CODE    SEGMENT
        ASSUME   CS:CODE
START:  MOV      AH,1
        INT      21H           ;等待键入字符,送 AL
```

```

        CMP     AL, 0DH           ;是不是回车符?
        JZ      DONE            ;是则转 DONE 退出程序
        CMP     AL, '0'
        JB      NEXT
        CMP     AL, '9'
        JA      CHARUP          ;如不是 0~9 则转入 CHARUP 执行
        MOV     DL, AL
        MOV     AH, 2
        INT     21H
        JMP     START
CHARUP:  CMP     AL, 'A'
        JB      NEXT
        CMP     AL, 'Z'
        JA      CHRDN          ;如不是大写字母,则转入 CHRDN 执行
DISPC:  MOV     DL, 'C'
        MOV     AH, 2
        INT     21H
NEXT:   JMP     START
CHRDN:  CMP     AL, 'a'
        JB      NEXT
        CMP     AL, 'z'
        JA      NEXT          ;如不是小写字母,则不显示,转回 START
        JMP     DISPC
DONE:   MOV     AH, 4CH
        INT     21H
CODE    ENDS
        END     START

```

思考题：本程序采用 JA 和 JB 来判断输入字符在哪个区间？请问 JA 和 JB 这两条指令是依据哪些标志位来判断的？它们的成立条件分别是什么？

扩展题：试编写 ASCII 码的查询程序。即从键盘输入一个字符后，使屏幕上显示该字符的 ASCII 码，要求以两位十六进制方式显示。

[程序 3] 设平面上一点 P 的直角坐标为 (X, Y) ， X, Y 为有符号数，试编制若 P 落在第 i 象限内，则令 $k=i$ ；若 P 落在坐标轴上，则令 $k=0$ 的程序。

程序清单如下：

```

DATA    SEGMENT
X        DW      8300H
Y        DW      8200H
K        DB      0
DATA    ENDS
CODE    SEGMENT
        ASSUME CS:CODE, DS:DATA
START:  MOV     AX, DATA

```

```

MOV     DS,AX
AND     X,0FFFFH           ;判断 X 是否为 0
JZ      K0
AND     Y,0FFFFH           ;判断 Y 是否为 0
JZ      K0
TEST    X,8000H             ;测试 X 符号
JZ      K4
TEST    Y,8000H             ;测试 Y 符号
JZ      K2
MOV     K,3
JMP     EXIT
K2:     MOV     K,2
JMP     EXIT
K0:     MOV     K,0
JMP     EXIT
K4:     TEST    Y,8000H
JZ      K1
MOV     K,4
JMP     EXIT
K1:     MOV     K,1
EXIT:   MOV     AH,4CH
INT     21H
CODE    ENDS
END     START

```

思考题：本题还可以通过判断 X 和 Y 同号还是异号来确定其在哪一个象限？请问程序应如何修改？

扩展题：编制程序,判断键盘输入的一个数,若输入 1~5,则将该数转换为二进制数存放至 RLT 单元,若输入其他数则显示输出“ERROR”。

3.3 循环程序设计

实验目的：

- (1) 加深对循环结构的理解。
- (2) 掌握循环程序的设计方法。
- (3) 熟练掌握 DEBUG 的常用命令,学会用 DEBUG 调试程序。

〔程序 1〕 已知以 BUF 为首地址的字存储区中存放着 8 个有符号二进制数,试编写程序将其中大于等于 0 的数依次送入 BUF1 为首地址的字存储区中,小于 0 的数依次送入以 BUF2 为首地址的字存储区中。同时将大于等于 0 的数的个数送入 A 字变量,将小于 0 的数的个数送入 B 字变量。

编写的程序清单如下：

```
DATA    SEGMENT
BUF      DW      23,123,-12,-210,45,0,90,-453
BUF1     DW      8   DUP(0)
BUF2     DW      8   DUP(0)
A        DW      0
B        DW      0
DATA     ENDS
CODE     SEGMENT
        ASSUME    CS:CODE,DS:DATA
START:   MOV      AX,DATA
        MOV      DS,AX
        LEA      BX,BUF
        LEA      SI,BUF1
        LEA      DI,BUF2
        MOV      A,0
        MOV      B,0
        MOV      CX,8                ;置循环初值
L0:      MOV      AX,[BX]              ;判断元素是否非负,非负则转 L1
        CMP      AX,0
        JGE      L1
        MOV      [DI],AX              ;为负则将元素送 BUF2,执行 B+1
        ADD      DI,2
        INC      B
        JMP      NEXT
L1:      MOV      [SI],AX              ;非负则将元素送 BUF1,执行 A+1
        ADD      SI,2
        INC      A
NEXT:    ADD      BX,2                ;修改 BUF 的地址
        LOOP     L0                  ;修改循环次数,未完则转 L0
        MOV      AH,4CH
        INT      21H
CODE     ENDS
        END      START
```

思考题：本程序属于循环次数已知的程序，使用 CX 寄存器存放计数值，使用 LOOP 指令完成循环。但是在有些情况下，是不可以用 LOOP 指令的，比如循环中要用到移位指令时需要占用 CL 寄存器。我们仍可以两条指令完成完全相同的功能，请问是哪两条指令？

扩展题：统计以 BUF 为首地址开始的字节数据块中的正数的个数，将统计结果放至 RLT 单元。

【程序 2】 设 STR 字符串是以 0 结尾。试编写一个把字符串中的所有大写字母改为小写字母的程序，并将转换后的字符串显示输出。

编写的程序清单如下：

```
DATA    SEGMENT
STR      DB      'HOW arE YoU! ',0
DATA    ENDS
CODE    SEGMENT
        ASSUME    CS:CODE,DS:DATA
START:  MOV      AX,DATA
        MOV      DS,AX
        MOV      SI,OFFSET STR          ;取字符串开始地址
AGAIN:  MOV      DL,[SI]                ;取一字符
        OR       DL,DL                  ;是否到字符串尾?
        JZ       OK                     ;到字符串尾,转 OK
        CMP      DL,'A'                 ;否则,判断是否为大写字母
        JB       NEXT                   ;否,转 NEXT
        CMP      DL,'Z'
        JA       NEXT                   ;否,转 NEXT
        ADD      DL,20H                  ;是大写字母,改为小写字母
        MOV      [SI],DL                ;送回到字符串中
NEXT:   MOV      AH,2
        INT      21H
        INC      SI                      ;调整指针
        JMP      AGAIN                  ;继续循环
OK:     MOV      AH,4CH
        INT      21H
CODE    ENDS
        END      START
```

思考题：如果程序要求把小写字母都改为大写字母,程序段该如何修改?

扩展题：有一系列以 \$ 为结束符的字符串,对其中的非数字字符计数,保存计数结果。

[程序 3] 从键盘输入一字符串(字符数>1),然后在下一行以相反的次序显示出来(采用 DOS 9 号和 10 号系统功能调用)。

本程序需要建立两个指针指向输入的字符串,一个指向串首,一个指向串尾。将两个指针指向的字符交换,字符交换的操作要进行到字符串首指针的值大于等于字符串尾指针的值为止,即可将字符串的次序颠倒。

程序清单如下：

```
DATA    SEGMENT
BUF      DB      255,0,255 DUP(0)
DATA    ENDS
CODE    SEGMENT
        ASSUME    CS:CODE,DS:DATA
START:  MOV      AX,DATA
```

```

MOV     DS,AX
LEA     DX,BUF                      ;输入一串字符
MOV     AH,10
INT     21H
MOV     SI,OFFSET BUF+2            ;SI 指向串首
MOV     BX,SI                      ;BX 指向串尾+1
ADD     BL,BUF+1                   ;串首偏移地址加上输入字符的个数
ADC     BH,0
MOV     BYTE PTR[BX], '$ '        ;串尾后送结束符
DEC     BX                        ;BX 指向串尾
AGAIN:  MOV    AL,[BX]              ;交换字符
XCHG    AL,[SI]
MOV     [BX],AL
DEC     BX                        ;指针调整
INC     SI
CMP     SI,BX                      ;指针比较
JC      AGAIN
MOV     BUF+1,0AH
MOV     DX,OFFSET BUF+1           ;从换行开始输出
MOV     AH,9
INT     21H
MOV     AH,4CH
INT     21H
CODE    ENDS
END     START

```

思考题：请画出本程序的流程图。

扩展题：从 STRN 地址开始有一字符串，以 ‘\$’ 作为结束标志，要求统计该字符串长度并存于 LENG 单元。

3.4 综合程序设计

[程序 1] 在内存 DATA 单元开始存放若干个无符号数，数据个数在 COUNT 单元存放。编制程序分别计算其中奇数、偶数的和，并分别存入 ODDSUM EVENSUM 单元。设各类和不超过 16 位二进制数，可用一个字表示或存放。

本例对每一个数据进行判断、累加都是进行重复的操作，因此可编制循环结构程序，以 COUNT 控制循环的次数。

主程序清单如下：

```

DSEG    SEGMENT
DATA     DB          15H,26H,03H,64H,8AH,0AAH,24H,48H
COUNT   DW          08

```

```

ODDSUM    DW      0
EVENSUM    DW      0
DSEG  ENDS
CSEG  SEGMENT
        ASSUME    CS:CSEG,DS:DSEG
MAIN:  MOV      AX,DSEG
        MOV      DS,AX
        LEA      SI,DATA          ;设置地址指针
        MOV      CX,COUNT        ;计数值送 CX
        XOR      AX,AX           ;AX 清零
        XOR      BX,BX           ;清零和寄存器
        XOR      DX,DX
AGAIN: MOV      AL,[SI]          ;取数据
        TEST     AL,01           ;测试最低位
        JZ       EVENS          ;偶数,转 EVENS
        ADD      BX,AX           ;奇数,累计和
        JMP      CHCNT
EVENS:  ADD      DX,AX           ;偶数,累计和
CHCNT:  INC      SI              ;指向下个数据
        LOOP     AGAIN          ;计算完? 未完继续
        MOV      ODDSUM,BX       ;保存结果
        MOV      EVENSUM,DX
        MOV      AH,4CH
        INT      21H
CSEG  ENDS
        END      MAIN

```

思考题：测试最低位状态也可以用右移指令，然后判断 CF 的状态。请问如何修改程序语句？

扩展题：有一个班级，学生人数存储于内存 NUB 单元，该班同学某门课程的成绩存放于内存 S 开始的单元，编制程序，统计该班该课程的平均成绩，存放于 E 单元。

〔程序 2〕 试编制一程序。从键盘输入一个字符，若为“1”，则显示“THE FIRST SUBROUTINE!”；否则显示“INPUT THE RIGHT CHARACTER”，返回 DOS。

本例用了三个非常常用的 DOS 系统功能调用。其中第一个调用实现一个字符串的输出显示，调用前，应将 DX 指向存放要显示字符串的缓冲区的首地址，被显示的字符串应以‘\$’结尾。第二个调用，实现从键盘输入一个字符，调用后，字符的 ASCII 码在 AL 寄存器中。第三个调用为在屏幕上显示一个字符，调用前，被显示字符的 ASCII 码应送入 DL 寄存器中。有关 DOS 系统调用的详细资料参见附表 A-IV。

整个程序的清单如下：

```

DATA  SEGMENT
STR0  DB      'INPUT THE RIGHT CHARACTER','$ '
STR1  DB      'INPUT CHARACTER:$ '

```